Maple Lab #2 for Math 211

Using the LinearAlgebra Package

> $restart; with(LinearAlgebra)$: #reads in the LinearAlgebra package

1. Defining Matrices and Vectors

- defining a matrix by listing its elements (three methods) and by a formula:

> $A := Matrix([[1, 2, 3], [-2, 4, 5]]); A1 := \langle\langle 1, 2, 3\rangle | \langle -2, 4, 5\rangle\rangle;$
> $B := Matrix(2, 3, [2, 2, 3, 0, 1, 1]);\ C := Matrix(3, 3, (i, j) \to i^j);$

$$A := \begin{bmatrix} 1 & 2 & 3 \\ -2 & 4 & 5 \end{bmatrix}$$

$$A1 := \begin{bmatrix} 1 & -2 \\ 2 & 4 \\ 3 & 5 \end{bmatrix}$$

$$B := \begin{bmatrix} 2 & 2 & 3 \\ 0 & 1 & 1 \end{bmatrix}$$

$$C := \begin{bmatrix} 1 & 1 & 1 \\ 2 & 4 & 8 \\ 3 & 9 & 27 \end{bmatrix} \qquad (1)$$

Note that these have type "Matrix"

> $type(A, Matrix), type(A1, Matrix), type(B, Matrix), type(C, Matrix);$

$$true, true, true, true \qquad (2)$$

- vectors are defined as follows:

> $w1 := Vector([1, -2]); w2 := \langle 2, 3\rangle; v := Vector(3, 1);$

$$w1 := \begin{bmatrix} 1 \\ -2 \end{bmatrix}$$

$$w2 := \begin{bmatrix} 2 \\ 3 \end{bmatrix}$$

$$v := \begin{bmatrix} 1 \\ 1 \\ 1 \end{bmatrix} \qquad (3)$$

-to find the number of rows/columns of a matrix and the number of entries of a vector, use

> $RowDimension(A), ColumnDimension(A);\ Dimensions(A),\ Dimension(v);$

$$2, 3$$

$$2, 3, 3 \qquad (4)$$

-special matrices: the 3 x 3 identity matrix, zero matrix, and a diagonal matrix:

> $IdentityMatrix(3), ZeroMatrix(3,3), DiagonalMatrix([1,3,5]);$

$$\begin{bmatrix}1&0&0\\0&1&0\\0&0&1\end{bmatrix}, \begin{bmatrix}0&0&0\\0&0&0\\0&0&0\end{bmatrix}, \begin{bmatrix}1&0&0\\0&3&0\\0&0&5\end{bmatrix} \tag{5}$$

2. Basic Matrix and Vector Operations:

-adding matrices (must have the same dimensions)

> $A + B;$

$$\begin{bmatrix}3&4&6\\-2&5&6\end{bmatrix} \tag{6}$$

- vector addition:

> $w1 + w2;$

$$\begin{bmatrix}3\\1\end{bmatrix} \tag{7}$$

-multiplying a matrix by a scalar (two methods):

> $2 \cdot A, ScalarMultiply(A, 2);$

$$\begin{bmatrix}2&4&6\\-4&8&10\end{bmatrix}, \begin{bmatrix}2&4&6\\-4&8&10\end{bmatrix} \tag{8}$$

-computing a linear combinations of vectors:

> $w := 3 \cdot w1 - 2 \cdot w2;$

$$w := \begin{bmatrix}-1\\-12\end{bmatrix} \tag{9}$$

-matrix multiplication (two methods):

> $A.C, Multiply(A, C);$

Note that we cannot multiply A by B since both are 2 x 3 matrices. However, we can multiply A by the transpose of B (on both sides):

> $Bt := Transpose(B);$

$$Bt := \begin{bmatrix}2&0\\2&1\\3&1\end{bmatrix} \tag{10}$$

> $Multiply(A, Bt), Multiply(Bt, A);$

$$\begin{bmatrix}15&5\\19&9\end{bmatrix}, \begin{bmatrix}2&4&6\\0&8&11\\1&10&14\end{bmatrix} \tag{11}$$

-powers (for square matrices only):

> $C^3;$

(12)

$$\begin{bmatrix} 142 & 386 & 1090 \\ 964 & 2644 & 7504 \\ 3078 & 8466 & 24066 \end{bmatrix} \quad (12)$$

- the inverse of a square matrix (if it's invertible):

> $C^{-1}, \frac{1}{C};$

$$\begin{bmatrix} 3 & -\frac{3}{2} & \frac{1}{3} \\ -\frac{5}{2} & 2 & -\frac{1}{2} \\ \frac{1}{2} & -\frac{1}{2} & \frac{1}{6} \end{bmatrix}, \begin{bmatrix} 3 & -\frac{3}{2} & \frac{1}{3} \\ -\frac{5}{2} & 2 & -\frac{1}{2} \\ \frac{1}{2} & -\frac{1}{2} & \frac{1}{6} \end{bmatrix} \quad (13)$$

- multiplying a matrix by a vector (three methods):

> $Cv := C\,.\,v;\ Multiply(C, v),\ MatrixVectorMultiply(C, v);$

$$Cv := \begin{bmatrix} 3 \\ 14 \\ 39 \end{bmatrix}$$

$$\begin{bmatrix} 3 \\ 14 \\ 39 \end{bmatrix}, \begin{bmatrix} 3 \\ 14 \\ 39 \end{bmatrix} \quad (14)$$

Note that v is viewed as a column vector. The result is a vector, not a matrix:

> $type(Cv, Vector),\ type(Cv, Matrix);$

$$true, false \quad (15)$$

- multiplying a vector by a matrix:

> $vC := Multiply(v, C);$

```
Error, (in Multiply) cannot multiply a column Vector and a Matrix
```

Note that in the LinearAlgebra package a vector is viewed only as a column vector. To multiply by a vector on the left, use the transpose:

> $vC := Transpose(v).C;$

$$vC := \begin{bmatrix} 6 & 14 & 36 \end{bmatrix} \quad (16)$$

- the dot product of two vectors (two methods):

> $DotProduct(w1, w2),\ w1\,.\,w2;$

$$-4, -4 \quad (17)$$

- the length (or norm) of a vector:

> $Norm(v, 2)$

$$\sqrt{3} \quad (18)$$

- the angle between two vectors:

> $VectorAngle(w1, -w2);$

$$\arccos\left(\frac{4\sqrt{5}\sqrt{13}}{65}\right) \tag{19}$$

For special angles, Maple will evaluate them exactly:

```
> VectorAngle(⟨3, 2⟩, ⟨-2, 3⟩), VectorAngle(⟨1, 0⟩, ⟨1, 1⟩),
  VectorAngle(⟨1, 0⟩, ⟨-1, sqrt(3)⟩);
```

$$\frac{\pi}{2}, \frac{\pi}{4}, \frac{2\pi}{3} \tag{20}$$

- to apply a function to each entry of a matrix, use map(..).

For example: to add 1 to each entry, or to square each entry, use the following:

```
> A, map(x→ x + 1, A), map(x→x^2, A);
```

$$\begin{bmatrix} 1 & 2 & 3 \\ -2 & 4 & 5 \end{bmatrix}, \begin{bmatrix} 2 & 3 & 4 \\ -1 & 5 & 6 \end{bmatrix}, \begin{bmatrix} 1 & 4 & 9 \\ 4 & 16 & 25 \end{bmatrix} \tag{21}$$

3. Extracting parts of a matrix:

- the (i,j)-th entry of A is given by A[i,j]:

```
> A[1, 2];
```

$$2 \tag{22}$$

This is the entry of A located in the 1st row, 2nd column.

- extracting a single row or column of a matrix (two methods):

```
> Row(A, 1), A[1, 1 ..-1], Column(A, 2), A[1 ..-1, 2]; # note that these are vectors, not matrices.
```

$$\begin{bmatrix} 1 & 2 & 3 \end{bmatrix}, \begin{bmatrix} 1 & 2 & 3 \end{bmatrix}, \begin{bmatrix} 2 \\ 4 \end{bmatrix}, \begin{bmatrix} 2 \\ 4 \end{bmatrix} \tag{23}$$

```
> type(Row(A, 1), Vector);
```

$$true \tag{24}$$

- extracting a submatrix consisting of various rows an columns of matrix (two methods):

```
> SubMatrix(B, 1 ..2, 1 ..2), B[1 ..2, 1 ..2]
      # the matrix consisting of the 1st two rows and columns of B
```

$$\begin{bmatrix} 2 & 2 \\ 0 & 1 \end{bmatrix}, \begin{bmatrix} 2 & 2 \\ 0 & 1 \end{bmatrix} \tag{25}$$

Here is the matrix C, and the submatrix obtained by taking the 1st and 3rd row and 1st and 3rd column (two methods) :

```
> C, C[[1, 3], [1, 3]], SubMatrix(C, [1, 3], [1, 3]);
```

$$\begin{bmatrix} 1 & 1 & 1 \\ 2 & 4 & 8 \\ 3 & 9 & 27 \end{bmatrix}, \begin{bmatrix} 1 & 1 \\ 3 & 27 \end{bmatrix}, \begin{bmatrix} 1 & 1 \\ 3 & 27 \end{bmatrix} \tag{26}$$

-changing an entry (in an existing matrix):

```
> B[1, 1] := 0; B;
```

$$B_{1,1} := 0$$

(27)

$$\begin{bmatrix} 0 & 2 & 3 \\ 0 & 1 & 1 \end{bmatrix} \quad (27)$$

4. Pasting Matrices

- put matrix A on top of matrix B:

> $S := \langle A, B \rangle;$

$$S := \begin{bmatrix} 1 & 2 & 3 \\ -2 & 4 & 5 \\ 0 & 2 & 3 \\ 0 & 1 & 1 \end{bmatrix} \quad (28)$$

-put matrix A to the left of matrix B:

> $AB := \langle A | B \rangle;$

$$AB := \begin{bmatrix} 1 & 2 & 3 & 0 & 2 & 3 \\ -2 & 4 & 5 & 0 & 1 & 1 \end{bmatrix} \quad (29)$$

-constructing block diagonal matrices:

> *DiagonalMatrix*([*C*, *C*, *IdentityMatrix*(2)]); *#note that the given matices must be square*

$$\begin{bmatrix} 1 & 1 & 1 & 0 & 0 & 0 & 0 & 0 \\ 2 & 4 & 8 & 0 & 0 & 0 & 0 & 0 \\ 3 & 9 & 27 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & 1 & 1 & 0 & 0 \\ 0 & 0 & 0 & 2 & 4 & 8 & 0 & 0 \\ 0 & 0 & 0 & 3 & 9 & 27 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 \end{bmatrix} \quad (30)$$

5. Row Reducing Matrices:

- to row reduce to an upper triangular matrix:

> *GaussianElimination*(*S*);

$$\begin{bmatrix} 1 & 2 & 3 \\ 0 & 8 & 11 \\ 0 & 0 & \frac{1}{4} \\ 0 & 0 & 0 \end{bmatrix} \quad (31)$$

Note that this doesn't explain how the row reduction was performed. If we want to see the individual steps, then we need to follow through the row reduction by ourselves by using the elementary row operation command RowOperation(..) . (Note that below the original S is listed for comparison purposes.)

> *S*, *RowOperation*(*S*, [2, 1], 3);
#the command addrow(..) replaces row 2 by 3·row 1 plus row2 in the matrix S

$$\begin{bmatrix} 1 & 2 & 3 \\ -2 & 4 & 5 \\ 0 & 2 & 3 \\ 0 & 1 & 1 \end{bmatrix}, \begin{bmatrix} 1 & 2 & 3 \\ 1 & 10 & 14 \\ 0 & 2 & 3 \\ 0 & 1 & 1 \end{bmatrix} \quad (32)$$

> *S*, *RowOperation*(*S*, [2, 4]); *# this interchanges rows 2 and 4 in S*

$$\begin{bmatrix} 1 & 2 & 3 \\ -2 & 4 & 5 \\ 0 & 2 & 3 \\ 0 & 1 & 1 \end{bmatrix}, \begin{bmatrix} 1 & 2 & 3 \\ 0 & 1 & 1 \\ 0 & 2 & 3 \\ -2 & 4 & 5 \end{bmatrix} \quad (33)$$

> *S*, *RowOperation*(*S*, 2, −4); *# this multiplies row 2 of S by -4*

$$\begin{bmatrix} 1 & 2 & 3 \\ -2 & 4 & 5 \\ 0 & 2 & 3 \\ 0 & 1 & 1 \end{bmatrix}, \begin{bmatrix} 1 & 2 & 3 \\ 8 & -16 & -20 \\ 0 & 2 & 3 \\ 0 & 1 & 1 \end{bmatrix} \quad (34)$$

- to find the reduced row echelon form (rref) of a matrix A, use:

> *ReducedRowEchelonForm*(*S*), *ReducedRowEchelonForm*(*AB*);

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \\ 0 & 0 & 0 \end{bmatrix}, \begin{bmatrix} 1 & 0 & \frac{1}{4} & 0 & \frac{3}{4} & \frac{5}{4} \\ 0 & 1 & \frac{11}{8} & 0 & \frac{5}{8} & \frac{7}{8} \end{bmatrix} \quad (35)$$

6. Solving a System of Linear Equations (given in matrix form):

- to solve a homogeneous system Ax = 0, we can use NullSpace(A):

> *NullSpace*(*AB*); *NullSpace*(*S*);

$$\left\{ \begin{bmatrix} -\frac{5}{4} \\ -\frac{7}{8} \\ 0 \\ 0 \\ 0 \\ 1 \end{bmatrix}, \begin{bmatrix} -\frac{3}{4} \\ -\frac{5}{8} \\ 0 \\ 0 \\ 1 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 0 \\ 0 \\ 1 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} -\frac{1}{4} \\ -\frac{11}{8} \\ 1 \\ 0 \\ 0 \\ 0 \end{bmatrix} \right\}$$

$$\varnothing \quad (36)$$

Note that this gives a basis of the nullspace/kernel of AB (respectively, of S). Note that Nullspace(S) = {0},
and that the basis of {0} is the empty set. In the first case, the general solution of ABx = 0 is:
t_1*[0,0,0,1,0,0] +... t_4*[1,0,0,0,7,-5] , where t_1, ..., t_4 in R.

This is made more explicit by using LinearSolve (except that Maple doesn't write it as a linear combination):

```
> sol := LinearSolve(AB, Vector(2, 0));
```

$$sol := \begin{bmatrix} \_t0_1 \\ \_t0_2 \\ \_t0_3 \\ \_t0_4 \\ 7\,\_t0_1 - 10\,\_t0_2 - 12\,\_t0_3 \\ -5\,\_t0_1 + 6\,\_t0_2 + 7\,\_t0_3 \end{bmatrix} \quad \textbf{(37)}$$

Here, the symbols _t[1], .., _t[4] are parameters. To find a specific solution, use subs (and map):

```
> map(x → subs(_t[1] = 1, _t[2] = 0, _t[3] = 0, _t[4] = 0, x), sol);
```

$$\begin{bmatrix} \_t0_1 \\ \_t0_2 \\ \_t0_3 \\ \_t0_4 \\ 7\,\_t0_1 - 10\,\_t0_2 - 12\,\_t0_3 \\ -5\,\_t0_1 + 6\,\_t0_2 + 7\,\_t0_3 \end{bmatrix} \quad \textbf{(38)}$$

```
> map(x → subs(_t[1] = 1, _t[2] = 1, _t[3] = 0, _t[4] = 1, x), sol);
```

$$\begin{bmatrix} 1 \\ 1 \\ 0 \\ 1 \\ -3 \\ 1 \end{bmatrix} \quad \textbf{(39)}$$

-to solve an inhomgeneous system Ax = b, use LinearSolve:

```
> LinearSolve(AB, w1);
> LinearSolve(S, Vector([1, 2, 3, 4]));
```

Here the linear system is inconsistent (has no solution).

7. Eigenvalues and Eigenvectors (for square matrices only)

```
> V3 := VandermondeMatrix([-1, 0, 1]); V4 := VandermondeMatrix([-1, 0, 1, 2]);
    # this defines a Vandermonde matrix
```

$$V3 := \begin{bmatrix} 1 & -1 & 1 \\ 1 & 0 & 0 \\ 1 & 1 & 1 \end{bmatrix}$$

$$V4 := \begin{bmatrix} 1 & -1 & 1 & -1 \\ 1 & 0 & 0 & 0 \\ 1 & 1 & 1 & 1 \\ 1 & 2 & 4 & 8 \end{bmatrix} \quad \textbf{(40)}$$

- the characteristic polynomial of a square matrix (here: of V3 and of V4)

> $f3 := CharacteristicPolynomial(V3, x); f4 := CharacteristicPolynomial(V4, x);$

$$f3 := x^3 - 2x^2 + x - 2$$

$$f4 := x^4 - 10x^3 + 14x^2 - 2x + 12 \quad \textbf{(41)}$$

> $Eigenvalues(V3);$

$$\begin{bmatrix} 2 \\ \mathrm{I} \\ -\mathrm{I} \end{bmatrix} \quad \textbf{(42)}$$

Thus, the eigenvalues of V3 are 2, I, -I. Recall that I is Maple's symbol for the complex number i (with i^2 = -1).
We can also find these by factoring f3 (over Q):

> $factor(f3);$

$$(x - 2)\left(x^2 + 1\right) \quad \textbf{(43)}$$

- **eigenvectors (and eigenvalues):**

> $EV3 := Eigenvectors(V3);$

$$EV3 := \begin{bmatrix} 2 \\ \mathrm{I} \\ -\mathrm{I} \end{bmatrix}, \begin{bmatrix} \frac{2}{3} & -1 & -1 \\ \frac{1}{3} & \mathrm{I} & -\mathrm{I} \\ 1 & 1 & 1 \end{bmatrix} \quad \textbf{(44)}$$

Note that this is a sequence consisting of 2 elements. The first is

> $EV3[1];$

This is a Vector consisting of the eigenvalues of A, and the second is a Matrix whose columns are of the eigenspace associated to the eigenvalue.

Note: The eigenvalues are listed according to their algebraic multiplicities:

> $J := DiagonalMatrix([JordanBlockMatrix([[3, 3]]), JordanBlockMatrix([[1, 2]])]);$
$Eigenvectors(J);$

$$J := \begin{bmatrix} 3 & 1 & 0 & 0 & 0 \\ 0 & 3 & 1 & 0 & 0 \\ 0 & 0 & 3 & 0 & 0 \\ 0 & 0 & 0 & 1 & 1 \\ 0 & 0 & 0 & 0 & 1 \end{bmatrix}$$

$$\begin{bmatrix} 1 \\ 1 \\ 3 \\ 3 \\ 3 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \end{bmatrix} \tag{45}$$

>

>